Daten gruppiert in Excel-Arbeitsmappen exportieren, Teil 1
Versionen: Access 2007, 2003, 2002/XP, 2000 und 97

Beim Export von Daten nach Excel sollen gruppierte Daten oftmals gemäß Gruppierungsfeld in verschiedene Tabellenblätter abgelegt werden. Zum Beispiel: Sie möchten Ihre Kundenadressen für eine Serienbriefaktion gruppiert nach Land in eine Arbeitsmappe exportieren, so dass anschließend Tabellen wie "Argentinien", "Belgien", "Chile", "Dänemark" usw. mit den entsprechenden Adressen vorhanden sind und gezielt im Seriendruck ausgewählt werden können. Mit den Bordmitteln von Access ist diese Aufgabe nicht zu erledigen, mit Hilfe einer VBA-Prozedur und Office Automation gelangen Sie dennoch zum Ziel:

1. Wechseln Sie mit Alt+F11 oder über das Menü Ansicht-Code in die VBA-Entwicklungsumgebung.

2. Legen Sie über das Menü Einfügen-Modul ein neues Modul "modGroupExportExcel" an.

3. Kopieren Sie die folgende Prozedur in das Modul:

Sub GroupExportToExcel(strSource As String, _
                       strGroupField As String, _
                       strWorkbook As String)
  Dim I As Long, J As Long
  Dim CurrRow As Long, CurrCol As Long
  Dim db As DAO.Database
  Dim rs As DAO.Recordset
  Dim objExcel As Excel.Application
  Dim wb As Excel.Workbook
  Dim ws As Excel.Worksheet
  Dim strGroup As String, strLast As String

  On Error Resume Next
  Set objExcel = CreateObject("Excel.Application")
  If Err <> 0 Or objExcel Is Nothing Then
    Beep
    MsgBox "Excel kann nicht gestartet " & _
           "werden: " & Err.Description, _
           vbOKOnly + vbCritical, "!!! Problem !!!"
    Exit Sub
  End If
  If strWorkbook = "" Then 'Leere Arbeitsmappe
    Set wb = objExcel.Workbooks.Add()
  Else 'Vorhandene Arbeitsmappe
    Set wb = objExcel.Workbooks.Open(strWorkbook)
  End If
  If Err <> 0 Or wb Is Nothing Then
    Beep
    MsgBox "Arbeitsmappe kann nicht " & _
           "angelegt/geöffnet werden: " & _
           Err.Description, _
           vbOKOnly + vbCritical, "!!! Problem !!!"
    Exit Sub
  End If

  DoCmd.Hourglass True
  Set db = CurrentDb()
  Set rs = db.OpenRecordset(strSource, _
                            dbOpenDynaset)
  strLast = ""
  Do While Not rs.EOF
    strGroup = rs(strGroupField)
    If strGroup <> strLast Then
      'Neue Gruppe/Tabelle
      strLast = strGroup
      Err = 0
      Set ws = wb.Sheets(strGroup)
      If Err <> 0 Then 'Nicht vorhanden, anlegen
        Set ws = _
            wb.Sheets.Add( _
            After:=wb.Sheets(wb.Sheets.Count))
        ws.Name = strGroup
      Else 'Vorhanden, Inhalt löschen
        ws.Cells.Select
        Selection.ClearContents
        ws.Range("A1").Select
      End If
      CurrCol = 1
      CurrRow = 1
      ws.Cells(CurrRow, CurrCol).Select
      For J = 0 To rs.Fields.Count - 1
        ws.Cells(CurrRow, CurrCol) = _
          rs(J).Name
        CurrCol = CurrCol + 1
      Next J
      CurrCol = 1
      CurrRow = CurrRow + 1
    End If 'Neue Gruppe?
    For J = 0 To rs.Fields.Count - 1
      ws.Cells(CurrRow, CurrCol).Value = _
        rs(J).Value
      CurrCol = CurrCol + 1
    Next J
    CurrCol = 1
    CurrRow = CurrRow + 1
    rs.MoveNext
  Loop

  rs.Close

  Set rs = Nothing
  Set db = Nothing

  objExcel.Visible = True

  Set objExcel = Nothing
  Set ws = Nothing
  Set wb = Nothing

  DoCmd.Hourglass False
  Beep

End Sub
· Speichern Sie die Änderungen.

Wie diese Prozedur funktioniert und wie Sie sie in der Praxis einsetzen, erläutern wir im zweiten Teil dieses Tipps in der nächsten Ausgabe.

Im ersten Teil dieses Tipps haben wir für den Export von gruppierten Daten in einzelne Tabellenblätter einer Excel-Arbeitsmappe die Basis in Form der Prozedur "GroupExportToExcel" geschaffen, die wie folgt deklariert war:

Sub GroupExportToExcel(strSource As String, _
                       strGroupField As String, _
                       strWorkbook As String)

Mit dem ersten Parameter "strSource" spezifizieren Sie eine Abfrage, die die zu exportierenden Daten liefert. Diese Abfrage muss so aufgebaut sein, dass die Datensätze in der richtigen Sortierung, also beispielsweise bei Kundenadressen nach Land und Firma oder bei einer Artikelverwaltung nach Warengruppe und Artikelname geliefert werden! Die Prozedur "GroupExportToExcel" nimmt keinerlei Gruppierung oder Sortierung vor, sondern verarbeitet die Datensätze in der gelieferten Reihenfolge!

Der zweite Parameter "strGroupField" gibt das Feld an, nach dem die Daten auf die Tabellenblätter verteilt werden. Wird hier zum Beispiel "Land" angegeben, werden Tabellenblätter wie "Argentinien", "Belgien", "Chile", "Dänemark" usw. mit den entsprechenden Adressen gefüllt, bei "Warengruppe" sind das Ergebnis Tabellenblätter wie zum Beispiel "Grafikkarten", "Netzteile", "Speicher", "Festplatten" usw., in denen die dazugehörenden Artikel aufgelistet sind.

Der dritte Parameter adressiert eine Excel-Arbeitsmappe, in der ggf. vorhandene Daten überschrieben und somit auf den aktuellsten Stand gebracht werden sollen. Wenn Sie hier eine leere Zeichenkette "" angeben, wird eine neue, leere Arbeitsmappe angelegt.

Der Aufruf der Prozedur könnte dann zum Beispiel so aussehen:

GroupExportToExcel "Kunden/Land", _
                   "Land", _
                   "Z:\Test\GroupExportExcel.xls"

Hier werden die von einer Abfrage "Kunden/Land" gelieferten Datensätze gemäß dem Feld "Land" in Tabellenblätter der Arbeitsmappe "Z:\Test\GroupExportExcel.xls" übertragen.

Die Funktionsweise der Prozedur "GroupExportToExcel" ist schnell erklärt: Im ersten Schritt wird per Office Automation (CreateObject) eine Verbindung zu Excel aufgebaut und dort ggf. die spezifizierte Arbeitsmappe geöffnet (Workbooks.Open) oder eine neue, leere Arbeitsmappe angelegt (Workbooks.Add). Danach wir die angegebene Abfrage als Recordset geöffnet und Datensatz für Datensatz durchlaufen. Anhand der aktuellen Gruppe in "strGroup" und einer zweiten Variablen "strLast" wird ein Gruppenwechsel erkannt. Ist das der Fall, wird zunächst versucht, auf ein eventuell vorhandenes Tabellenblatt zuzugreifen (ws = wb.Sheet(strGroup)) - schlägt das fehl, erfolgt die Neuanlage eines Tabellenblattes (wb.Sheets.Add) mit dem Namen der aktuellen Gruppe (ws.Name = strGroup). Ist das Tabellenblatt vorhanden, werden die Inhalte markiert und gelöscht, damit sie im folgenden durch aktuelle Daten ersetzt werden können. Dabei schreibt die Routine zunächst die Feldname in die erste Zeile des Tabellenblattes (Cells(Zeile,Spalte) = rs.Fields(x).Name) und danach bis zum nächsten Gruppenwechsel die dazugehörenden Datensätze (Cells(Zeile,Spalte) = rs.Fields(x).Value). Sind alle Datensätze durchlaufen, dereferenzieren wir die verwendeten Objektvariablen und geben einen akustischen Hinweis aus. Gleichzeitig wird Excel aktiviert und in den Vordergrund geholt, so dass die Daten direkt geprüft werden können.

